

ANÁLISIS DE ESTABILIDAD CON SPICE

J. I. Artigas, L. A. Barragán, J. M. Burdío y A. Sanz
Área de Tecnología Electrónica
Dpto. Ingeniería Eléctrica, Electrónica y Comunicaciones
María de Luna 3, 50015 ZARAGOZA. Tel. 76 761936. Fax. 76 762111
Universidad de Zaragoza
jiartigas@mcps.unizar.es

RESUMEN.- Este trabajo tiene como objetivo introducir al alumno en las técnicas de análisis de estabilidad de circuitos realimentados mediante SPICE. La aplicación de estas técnicas se muestra con varios ejemplos. Este trabajo está diseñado para formar parte de las prácticas de laboratorio de un curso de electrónica analógica.

1.- INTRODUCCIÓN

En el análisis de la estabilidad de circuitos realimentados es necesario conocer la ganancia en lazo abierto T del sistema en función de la frecuencia. A partir del diagrama de Bode de T se puede obtener si el circuito es estable o inestable, y caso de ser estable cuantificar la estabilidad mediante el margen de fase y de ganancia.

El parámetro T , conocido el modelo del amplificador, puede obtenerse teóricamente. Para ello, se ha de determinar el tipo de realimentación del circuito y calcular el factor de realimentación K . Estos cálculos, en cuanto el circuito presenta cierta dificultad, son largos, tediosos y propensos al error. Es interesante dotar al alumno de técnicas para obtener la ganancia en lazo abierto utilizando simuladores eléctricos tipo SPICE [1].

Las técnicas basadas en abrir el lazo de realimentación son de difícil aplicación pues pueden alterar la polarización o llevar a la saturación por el *offset* de entrada de los componentes. Además al abrir el lazo modificamos las impedancias de carga, lo que puede alterar los resultados.

En este trabajo se presentan dos métodos para determinar la estabilidad de circuitos electrónicos realimentados, que no presentan los inconvenientes de las técnicas basadas en abrir el lazo de realimentación.

2.- MÉTODOS DE ANÁLISIS

A continuación se presentan dos métodos para obtener la ganancia en lazo abierto T de circuitos realimentados. El primero de ellos, que denominaremos método A, es un clásico propuesto en la bibliografía que tiene un rango de aplicación general y es muy simple de aplicar. El segundo (método B) es un método cuya justificación es más intuitiva para el alumno pero cuya aplicación no es tan directa.

2.1.- Método A

El método desarrollado en [2]-[3] duplica el sistema a analizar. En la primera “copia” del sistema introduce una fuente de corriente AC en cualquier punto del bucle de realimentación (Ver Figura 1) y mide una ganancia en corriente del bucle T_i , donde:

$$T_i = \frac{i_y}{i_x} \quad (1)$$

en la segunda “copia” del circuito a analizar se inserta una fuente de voltaje AC en el lazo de realimentación (Ver Figura 1) y mide una ganancia en voltaje del bucle T_v , donde:

$$T_v = \frac{v_y}{v_x} \quad (2)$$

Estas medidas T_i y T_v , en general, difieren de la ganancia en bucle abierto T . Sin embargo, esta última, puede ser obtenida a partir de las dos primeras [3], a través de la siguiente fórmula:

$$T = \frac{T_i \cdot T_v - 1}{T_i + T_v + 2} \quad (3)$$




Figura 1: Medida de T_i y T_v

```

***** CIRCUITO A COMPROBAR: test *****
.subckt test left right
  * Aqui va el circuito a estudiar, sin fuentes y con el lazo abierto.
  * Los terminales del lazo se llaman left y right.
.ends

***** METODO A PARA ANALISIS DE ESTABILIDAD *****
xi Ti_left Ti_right test  ;* Componente test para medir Ti *
xv Tv_left Tv_right test  ;* Componente test para medir Tv *
*** Realiza la medida de Ti
iz 0 n1 AC 1
viy n1 Ti_left DC 0      ;* vix es una f. de tension para medir i *
vix n1 Ti_right DC 0     ;* viy es una f. de tension para medir i *
*** Realiza la medida de Tv
vz Tv_right Tv_left AC 1
evy vy 0 (0, Tv_left) 1  ;* vx y vy son dos nudos del circuito *
rvy vy 0 1G
evx vx 0 (Tv_right, 0) 1
rvx vx 0 1G
*****

.AC dec 50 1 10Meg      ;* ANALISIS AC *
.end

```

Figura 2: Aplicación del método A con SPICE

En la Figura 2 podemos ver las definiciones necesarias para realizar las simulaciones con SPICE de un circuito y obtener la ganancia en lazo abierto T y analizar su estabilidad. El circuito a simular se introduce dentro del subcircuito *test* del que se incluyen dos: uno llamado *xi* para medir T_i y otro denominado *xv* para medir T_v . El cálculo de T se puede realizar con las siguientes macros dentro de *probe*:

$$\begin{aligned} T_i &= i(viy) / i(vix) \\ T_v &= v(vy) / v(vx) \\ T &= (T_i \cdot T_v - 1) / (T_i + T_v + 2) \end{aligned}$$

2.2.- Método B

Se basa en la asimilación del circuito a analizar al esquema clásico de un sistema realimentado [4], como se muestra en la Figura 3. A partir de dicho esquema se obtienen las ecuaciones :

$$\begin{aligned} u_o &= A \cdot u_e \\ u_e &= u_i - K \cdot u_o \end{aligned} \quad (4)$$

La ganancia en lazo cerrado G se calcula como

$$G = \frac{u_o}{u_i} = \frac{A}{1+T} \quad (5)$$

siendo T la ganancia en lazo abierto

$$T = \frac{u_f}{u_e} = A \cdot K \quad (6)$$

Es posible realizar de forma directa el cálculo de T a partir de (5), del siguiente modo:

$$T = \frac{A}{G} - 1 \quad (7)$$

pudiendo medirse G y A directamente del circuito mediante PSPICE [1], teniendo en cuenta:

$$G = \frac{u_o}{u_i}, \quad A = \frac{u_o}{u_e} \quad (8)$$

Como la ganancia en lazo abierto depende de la ganancia del amplificador, en ambos métodos, los resultados de estabilidad serán tanto más correctos cuanto más exactos sean los modelos SPICE de los amplificadores con los que se trabaje.




Figura 3: Esquema de sistema realimentado.




Figura 4: Circuito del Ejemplo 1

```

*Ejemplo 1
*Modelo del opamp
.subckt opamp non inv out
rin non inv 1MEG
egain 1 0 (non , inv) 100K
ropen 1 2 1K
copen 2 0 15.92u
eout 3 0 (2,0) 1
rout 3 out 50
.ends

*Descripcion del circuito
Vi n 0 AC 1
X1 n i o opamp
R1 o 0 200
R2 o i 10K
R3 i 0 1K
C1 i 0 0.038u
.ac dec 101 1 1Meg
.end

```

Figura 5: Listado SPICE del Ejemplo 1

3.- RESULTADOS

A continuación se calcula la ganancia en lazo abierto de circuitos realimentados con amplificador operacional utilizando los dos métodos expuestos en la sección anterior. En todos los ejemplos, se ha utilizado un modelo de amplificador operacional que consta de un único polo.

3.1.- Método A

El esquema circuital del primer ejemplo, analizado en [3], se muestra en la Figura 4 y su descripción SPICE en la Figura 5.

Aplicando el método A para determinar la ganancia en lazo abierto T se obtienen los resultados simulados de amplitud y fase que se representan en las Figuras 6 y 7 que incluyen también las variables intermedias Ti y Tv necesarias para determinar T.

3.2.- Método B

Aplicando el método B para determinar la ganancia en lazo abierto T, se obtienen los resultados simulados de amplitud y fase que se representan en la Figura 8.




Figura 6: Amplitudes de T, Ti, Tv para el Ejemplo 1 con el método A




Figura 7: Fases de T, Ti, Tv para el Ejemplo 1 con el método A




Figura 8: Amplitud y fase de T para el Ejemplo 1 con el método B




Figura 9: Circuito del Ejemplo 2

La expresión de T, implementada con una macro, para el caso concreto del circuito de la Figura 4, es la siguiente:

$$T = ((V(o) / (V(n) - V(i)) / (V(o) / V(n))) - 1)$$

Se observa la coincidencia de resultados con el método anterior, sin necesidad de definir parámetros intermedios. A partir de la amplitud y fase de T, el alumno puede comprobar que el circuito es estable y medir los márgenes de fase y de ganancia.

Asimismo, el alumno puede observar el efecto que sobre la estabilidad tiene la introducción de nuevos elementos en el circuito o la modificación de sus valores. Este último caso se muestra a continuación con un segundo ejemplo [5] cuyo esquema circuital se representa en la Figura 9.

En la Figura 10 se representan los resultados simulados de amplitud y fase de T, del circuito sin el condensador C2. El margen de fase resulta ser 18°. Para mejorar la estabilidad se introduce C2, con lo que se aumenta el margen de fase a 49° (Figura 11).




Figura 10: Amplitud y fase de T para el Ejemplo 2 sin C2, con el método B




Figura 11: Amplitud y fase de T para el Ejemplo 2 con C2, con el método B

4.- CONCLUSIONES

Se pretende dotar al alumno de herramientas para analizar la estabilidad de circuitos mediante ordenador. En este trabajo se han utilizado dos métodos para ilustrar el problema de la estabilidad en frecuencia de etapas con amplificadores operacionales. En particular, el método A se ha mostrado muy sencillo de utilizar pues no requiere analizar el tipo de realimentación ni identificar el sistema. Únicamente se ha de introducir el circuito a analizar dentro del subcircuito de la Figura 2, se simula y se representa la amplitud y la fase de la ganancia en bucle abierto T.

Se ha utilizado el método A para realizar una práctica de la asignatura de Electrónica Analógica de Ingeniería Industrial. Durante la práctica el alumno visualiza la ganancia en bucle abierto de dos amplificadores operacionales: el 714, un amplificador compensado y el 748 un amplificador subcompensado. Se observa la respuesta temporal ante un impulso de una etapa estable y otra inestable, y por último se aplica el método A indicado en esta comunicación para mostrar al alumno el problema de la compensación en frecuencia.

5.- REFERENCIAS

- [1] MicroSim Corporation. "Manual de PSpice: Versión 5.0", 1991.
- [2] Middlebrook, R. D. *International Journal of Electronic*, vol. 38, nº 4, pp. 485-512, 1975.
- [3] Tuinenga, P. W. "SPICE: A Guide to Circuit Simulation and Analysis using PSpice". Prentice-Hall, 1992
- [4] Ogata, K. "Ingeniería de Control Moderna". Prentice-Hall, 1980.
- [5] M. Nurczyk. "Oscillators Don't Amplifiers Do!". *The computer Applications Journal*, pp. 20-25, Agosto 1993.º